# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=python-gpsoauth
_pkgname=gpsoauth
pkgver=2.0.0
pkgrel=1
pkgdesc='A python client library for Google Play Services OAuth.'
arch=('any')
url='https://pypi.org/project/gpsoauth/#files'
license=('MIT')
depends=('python' 'python-requests' 'python-pycryptodomex')
makedepends=('python-build' 'python-pip')
source=("https://files.pythonhosted.org/packages/source/${_pkgname::1}/$_pkgname/$_pkgname-$pkgver.tar.gz")
sha512sums=('e65daba3e96ff98693c0ba1d1af578d26feb875016592b4d9e5d03ebbdebb3a4eedfc1ce3d27597d2364113d5e9e3408098c2ac36369742ceb470dcd0c5edd2b')

build() {
  cd "$_pkgname-$pkgver"

  python -m build --wheel --outdir="$startdir/dist"
}

package() {
  cd "$_pkgname-$pkgver"

	  pip install \
    --verbose \
    --disable-pip-version-check \
    --no-warn-script-location \
    --ignore-installed \
    --no-compile \
    --no-deps \
    --root="$pkgdir" \
    --prefix=/usr \
    --no-index \
    --find-links="file://$startdir/dist" \
    $_pkgname
}

